9. Non-linear Algebra and Non-Linear Equations

2016-05-23

```
$Version
10.0 for Mac OS X x86 (64-bit) (September 10, 2014)
```

9.1 Finding Characteristic Quantities of Algebraic Expressions

Coefficient[*expr*, *form*]	coefficient of ***form*** in ***expr***
CoefficientList[poly, *form*]	give a list of the coefficients of powers of ***form*** in polynomial *poly*
CoefficientList[poly,*{var1 var2,... }*]	give an array of coefficients of var_i
Exponent[*expr*]	maximum power of ***form*** in ***expr***
Part[*expr*, n]	n-th term of ***expr***
Numerator[*expr*]	numerator (German: Zähler) of ***expr***
Denominator[*expr*]	denominator (German: Nenner) of ***expr***

```
Clear[e, x, y]

e = Expand[ (1 + 3 x + 4 y^2 )^2]
```

$1+6x+9x^2+8y^2+24xy^2+16y^4$

```
Coefficient[e, x]
```

$6+24y^2$

```
Coefficient[e, y]
```

0

```
Coefficient[e, y^2]
```

$8+24x$

```
Exponent[e,y]
```

4

```
pp = 17 + 2 x y^2 + 3 x^3 y
```

$17+3x^3y+2xy^2$

```
CoefficientList[pp, x]
```

$\{17, 2y^2, 0, 3y\}$

```
CoefficientList[pp, y]
```

$\{17, 3x^3, 2x\}$

```
CoefficientList[pp, {x, y}]
```

{{17, 0, 0}, {0, 0, 2}, {0, 0, 0}, {0, 3, 0}}

```
CoefficientList[pp, {y, x}]
```

{{17, 0, 0, 0}, {0, 0, 0, 3}, {0, 2, 0, 0}}

The dimensions of the array returned by CoefficientList are determined by the values of **Exponen-t[poly, vari].**

```
Exponent[pp, x]
```

3

```
Exponent[pp, y]
```

2

```
Part[e,5]
```

$24\, x\, y^2$

```
f = 2/3
```

$\frac{2}{3}$

```
Denominator[f]
```

3

```
Numerator[f]
```

2

```
s
```

s

```
Numerator[s]
```

s

```
Denominator[s]
```

1

9.2 Algebraic Operations on Polynomials

```
Clear[x, p1, p2 , q, r]
p1(x) = p2(x) * q(x)  +  r(x)
```

PolynomialQuotient[*poly1, poly2, x*]	find the result of dividing ***poly1*** in ***x*** by ***poly2***, dropping any remainder term
PolynomialRemainder[*poly1, poly2, x*]	find the remainder of dividing ***poly1*** in ***x*** by ***poly2***
Factor[*poly*]	factor the polynomial ***poly***

```
p1 = x^3 - 1
```

$-1 + x^3$

```
p2 = x - 1
```

$-1 + x$

```
PolynomialQuotient[p1, p2, x]
```

$1 + x + x^2$

```
PolynomialRemainder[p1, p2, x]
```

0

```
Factor[p1]
```

$(-1 + x)\ (1 + x + x^2)$

```
p1 = x^3 - 2
```

$-2 + x^3$

```
Factor[p1]
```

$-2 + x^3$

```
PolynomialQuotient[p1, p2, x]
```

$1 + x + x^2$

```
PolynomialRemainder[p1, p2, x]
```

-1

```
PolynomialRemainder[x^2, x + 1, x]
```

1

```
PolynomialQuotient[x^2, x + 1, x]
```

$-1 + x$

```
{PolynomialRemainder[x + y, x - y, x],
 PolynomialRemainder[x + y, x - y, y]  }
```

$\{2\ y,\ 2\ x\}$

```
p1 = (x + 2)(x + 1)(x + 3); p1 = Expand[p1]
```

$6 + 11\ x + 6\ x^2 + x^3$

```
p2 = (x + 3)(x + 2)(x - 1); p2 = Expand[p2]
```

$-6 + x + 4\ x^2 + x^3$

PolynomialGCD[*poly1, poly2*]	find the greatest common divisor of two polynomials
PolynomialLCM[*poly1, poly2*]	find the least common multiple of two polynomials

```
PolynomialGCD[p1, p2]//Factor
```

$(2 + x)\ (3 + x)$

```
PolynomialLCM[p1, p2]//Factor
```

$(-1 + x)\ (1 + x)\ (2 + x)\ (3 + x)$

Resultant[*poly1, poly2*]	find the resultant of two polynomials; this is zero if these have (a) common root(s)

p1 = $\left(\sqrt{2} - x\right)\ (x + 1)\ \left(x + \sqrt{2}\right)^2$**; p1 = Expand[p1]**

$2\sqrt{2} + 2\ x + 2\sqrt{2}\ x + 2\ x^2 - \sqrt{2}\ x^2 - x^3 - \sqrt{2}\ x^3 - x^4$

```
Clear[a]
p2 = (x^3 + 3 x^2 + a) ; p2 = Expand[p2]
```

$a + 3 x^2 + x^3$

```
Solve[p1 == 0, x]
```

$\{\{x \to -1\}, \{x \to -\sqrt{2}\}, \{x \to -\sqrt{2}\}, \{x \to \sqrt{2}\}\}$

```
Solve[p2 == 0, x]
```

$$\left\{\left\{x \to -1 + \frac{2^{1/3}}{\left(-2 - a + \sqrt{4 a + a^2}\right)^{1/3}} + \frac{\left(-2 - a + \sqrt{4 a + a^2}\right)^{1/3}}{2^{1/3}}\right\},\right.$$

$$\left\{x \to -1 - \frac{1 + i \sqrt{3}}{2^{2/3} \left(-2 - a + \sqrt{4 a + a^2}\right)^{1/3}} - \frac{\left(1 - i \sqrt{3}\right) \left(-2 - a + \sqrt{4 a + a^2}\right)^{1/3}}{2 \times 2^{1/3}}\right\},$$

$$\left.\left\{x \to -1 - \frac{1 - i \sqrt{3}}{2^{2/3} \left(-2 - a + \sqrt{4 a + a^2}\right)^{1/3}} - \frac{\left(1 + i \sqrt{3}\right) \left(-2 - a + \sqrt{4 a + a^2}\right)^{1/3}}{2 \times 2^{1/3}}\right\}\right\}$$

For which values of the variable a can the polynomials **p1(x)** and **p2(x)** have a common root ?

```
Resultant[p1, p2, x]
```

$-336 + 112 \sqrt{2} - 368 a + 104 \sqrt{2} a - 136 a^2 + 28 \sqrt{2} a^2 - 20 a^3 + 2 \sqrt{2} a^3 - a^4$

```
so = Solve[% == 0, a] // Union
```

$\{\{a \to -2\}, \{a \to 2 (-3 - \sqrt{2})\}, \{a \to 2 (-3 + \sqrt{2})\}\}$

```
p11 = p1 /. so[[1]] // Factor
p21 = p2 /. so[[1]] // Factor
```

$-(1 + x) (\sqrt{2} + x) (-2 + x^2)$

$(1 + x) (-2 + 2 x + x^2)$

a = - 2: -1 is a common root of p1 and p2.

```
p11 = p1 /. so[[2]] // Factor
p21 = p2 /. so[[2]] // Factor
Solve[p21 == 0]
```

$-(1 + x) (\sqrt{2} + x) (-2 + x^2)$

$-6 - 2 \sqrt{2} + 3 x^2 + x^3$

$\left\{\{x \to \sqrt{2}\}, \left\{x \to \frac{1}{2} \left(-3 - \sqrt{2} - i \sqrt{-3 + 6 \sqrt{2}}\right)\right\}, \left\{x \to \frac{1}{2} \left(-3 - \sqrt{2} + i \sqrt{-3 + 6 \sqrt{2}}\right)\right\}\right\}$

$a = 2 (-3 - \sqrt{2})$: $\sqrt{2}$ is a common root of p1 and p2.

```
p11 = p1 /. so[[3]] // Factor
p21 = p2 /. so[[3]] // Factor
Solve[p21 == 0]
```

$-(1 + x) (\sqrt{2} + x) (-2 + x^2)$

$-6 + 2 \sqrt{2} + 3 x^2 + x^3$

$\left\{\{x \to -\sqrt{2}\}, \left\{x \to \frac{1}{2} \left(-3 + \sqrt{2} - \sqrt{3 + 6 \sqrt{2}}\right)\right\}, \left\{x \to \frac{1}{2} \left(-3 + \sqrt{2} + \sqrt{3 + 6 \sqrt{2}}\right)\right\}\right\}$

$a = 2 (-3 + \sqrt{2}$ **:** $-\sqrt{2}$ is a common root of p1 and p2.

Often it is impossible to compute the roots of complicated polynomials. Even in such cases the **Resultant[]** tells us whether two polynomials have common roots. It assumes the value zero if there are common roots.

```
Resultant[ (x - y)^2  - 2, y^2 - 3, y]
```

$1 - 10\, x^2 + x^4$

```
Resultant[ (x - y)^2  - 2, y^2 - 3, x]
```

$\left(-3 + y^2\right)^2$

9.2.1 Field Extensions

In higher algebra a set $\mathcal{K}$ of numbers or/and symbols is called a field (German: Körper) if

1. it is closed under addition and multiplication,
2. is associative and commutative under both operations;
3. contains a unique unit element e;
4. contains the solution x of any equation a + x = b, with arbitrary a, b $\in \mathcal{K}$,
5. contains the solution x of any equation a x = b, with arbitrary a ≠ 0, b $\in \mathcal{K}$.

Examples are:

$\mathbb{Q}$, the field of all rational numbers p/q, with p, q (q ≠ 0) any integers.

$\mathbb{R}$, the field of all real numbers.

$\mathbb{C}$ the field of all complex numbers = $\mathbb{R}[i]$ with $i = \sqrt{-1}$, the imaginary unit.

It may well be possible that a non-linear equation cannot be solved over a field. For example, take the equation:

$$x^2 + 1 = 0$$

over the field $\mathcal{K} = \mathbb{R}$, the real numbers. This does not contain the imaginary unit i; so the above equation cannot be solved within $\mathbb{R}$. By joining the imaginary unit i to $\mathbb{Q}$ or $\mathbb{R}$, these fields are extended to that of the Gaussian rational field (= rational complex numbers) $\mathbb{Q}[i]$ or the field of all complex numbers $\mathbb{C} = \mathbb{R}[i]$. The above equation has the two solutions $\pm i$ within $\mathbb{C} = \mathbb{R}[i]$. Such an addition is called a **field extension.**

Similarly, the equation:

$$x^2 - 2 = 0$$

has no solutions over the field of fractions of integers $\mathcal{K} = \{a/b,\ b \neq 0,\ a, b \in \mathbb{Z}\} = \mathbb{Q}$, but it has the solutions $\pm\sqrt{2}$ over the field $\mathcal{K} = \mathbb{Q}[\sqrt{2}\,]$, So the above quadratic polynomial does not factor over $\mathcal{K} = \mathbb{Q}$, but it does factor over

$$\mathcal{K} = \mathbb{Q}[\sqrt{2}\,]: \qquad x^2 - 2 = \left(x - \sqrt{2}\right)\left(x + \sqrt{2}\right).$$

Such an extension can be implemented in *Mathematica* in two ways:

9.2.2.1 Using the option Extension

Factor[*poly*, Extenssion -> {*a*₁,*a*₂,....}]	factors ***poly*** over the field $\mathbb{Q}[a_1,\ a_2,\ \ldots.]$

```
Factor[x² - 2]
```

$-2 + x^2$

```
Factor[x² - 2, Extension → {√2}]
```

$-\left(\sqrt{2} - x\right)\left(\sqrt{2} + x\right)$

```
Factor[x^2 + 2, Extension → {ⅈ, √2}]
```

$\left(\sqrt{2} - i\,x\right)\left(\sqrt{2} + i\,x\right)$

```
Factor[3 x^2 - 2 y^2, Extension → {√2, √3}]
```

$-\frac{1}{3}\left(-3\,x + \sqrt{6}\,y\right)\left(3\,x + \sqrt{6}\,y\right)$

Now a cubic root is considered: $\alpha = \sqrt[3]{2}$, the real solution of the following polynomial:

```
ff = Factor[x^3 - 2, Extension → {∛2}]
```

$-\left(2^{1/3} - x\right)\left(2^{2/3} + 2^{1/3}\,x + x^2\right)$

```
so = Solve[ff[[3]] == 0]
```

$\left\{\left\{x \to \frac{1}{2}\left(-2^{1/3} - i\,2^{1/3}\sqrt{3}\right)\right\}, \left\{x \to \frac{1}{2}\left(-2^{1/3} + i\,2^{1/3}\sqrt{3}\right)\right\}\right\}$

```
Factor[x^3 - 2, Extension → {ⅈ , √3 , ∛2}]
```

$-\frac{1}{4}\left(2^{1/3} - x\right)\left(-i\,2^{1/3} + 2^{1/3}\sqrt{3} - 2\,i\,x\right)\left(i\,2^{1/3} + 2^{1/3}\sqrt{3} + 2\,i\,x\right)$

9.2.1.2 By joining a polynomial and using PolynomialRemainder

```
Clear[α]
pa = α^3 - 2;
```

In the extended field $\mathbb{Q}[\alpha]$ $\{a + b\,\alpha + c\,\alpha^2,\ a, b, c \in \mathbb{Q}\}$ any number f(a + b α + cα^2) (f(x) a polynomial operator of x) may be given as $a1 + b1\,\alpha + c1\,\alpha^2$ with a1, b1, c1 $\in$ $\mathbb{Q}$, Higher powers of α resulting from operations involving several such numbers are remomved with the help of PolynomialRemainder.
For example, the **product** $a + b\,\alpha + c\,\alpha^2$ of two such numbers $a1 + b1\,\alpha + c1\,\alpha^2$ and $a2 + b2\,\alpha + c2\,\alpha^2$
is expressed as:

$$(a1 + b1\,\alpha + c1\,\alpha^2)\,(a2 + b2\,\alpha + c2\,\alpha^2) = a + b\,\alpha + c\,\alpha^2\,.$$

The coefficients a, b, c representing the product $a + b\,\alpha + c\,\alpha^2$ of the two numbers can be expressed by a1, a2, b1, b2, c1, c2 as given below in **sopr.** The corresponding formulae are found in the following way:

```
pr = (a1 + b1 α + c1 α^2) (a2 + b2 α + c2 α^2) // Expand
```

$a1\,a2 + a2\,b1\,\alpha + a1\,b2\,\alpha + b1\,b2\,\alpha^2 + a2\,c1\,\alpha^2 + a1\,b2\,c2\,\alpha^2 + b2\,c1\,\alpha^3 + b1\,b2\,c2\,\alpha^3 + b2\,c1\,c2\,\alpha^4$

```
PolynomialRemainder[pr, pa, α ] - (a + b α + c α^2)
```

$-a + a1\,a2 + 2\,b2\,c1 + 2\,b1\,b2\,c2 - b\,\alpha + (a2\,b1 + a1\,b2 + 2\,b2\,c1\,c2)\,\alpha - c\,\alpha^2 + (b1\,b2 + a2\,c1 + a1\,b2\,c2)\,\alpha^2$

```
Thread[CoefficientList[%, α] == Table[0, {Exponent[pa, α]}]]
```

$\{-a + a1\,a2 + 2\,b2\,c1 + 2\,b1\,b2\,c2 == 0,\ -b + a2\,b1 + a1\,b2 + 2\,b2\,c1\,c2 == 0,\ b1\,b2 - c + a2\,c1 + a1\,b2\,c2 == 0\}$

```
sopr = Solve[%, {a, b, c}] // Flatten
```

$\{a \to a1\,a2 + 2\,b2\,c1 + 2\,b1\,b2\,c2,\ b \to a2\,b1 + a1\,b2 + 2\,b2\,c1\,c2,\ c \to b1\,b2 + a2\,c1 + a1\,b2\,c2\}$

```
pr = (a + b α + c α^2) /. sopr
```

$a1\,a2 + 2\,b2\,c1 + 2\,b1\,b2\,c2 + (a2\,b1 + a1\,b2 + 2\,b2\,c1\,c2)\,\alpha + (b1\,b2 + a2\,c1 + a1\,b2\,c2)\,\alpha^2$

The **inverse element** inv of na is found in the following way:

```
Clear[a, b, c, a1, b1, c1]
na = a1 + b1 α + c1 α^2 ;
```

```
inv = a + b α + c α^2 ;
```

na.inv = 1 = $(a1 + b1\,\alpha + c1\,\alpha^2)\,(a + b\,\alpha + c\,\alpha^2)$

```
CoefficientList[
   PolynomialRemainder[(a1 + b1 α + c1 α^2) (a + b α + c α^2), pa, α], α] == {1, 0, 0};
```

```
so = Solve[%, {a, b, c}] // Flatten
```

$$\left\{a \to -\frac{-a1^2 + 2\,b1\,c1}{a1^3 + 2\,b1^3 - 6\,a1\,b1\,c1 + 4\,c1^3},\ b \to -\frac{a1\,b1 - 2\,c1^2}{a1^3 + 2\,b1^3 - 6\,a1\,b1\,c1 + 4\,c1^3},\ c \to -\frac{-b1^2 + a1\,c1}{a1^3 + 2\,b1^3 - 6\,a1\,b1\,c1 + 4\,c1^3}\right\}$$

```
inv = a + b α + c α^2 /. so
```

$$-\frac{-a1^2 + 2\,b1\,c1}{a1^3 + 2\,b1^3 - 6\,a1\,b1\,c1 + 4\,c1^3} - \frac{(a1\,b1 - 2\,c1^2)\,\alpha}{a1^3 + 2\,b1^3 - 6\,a1\,b1\,c1 + 4\,c1^3} - \frac{(-b1^2 + a1\,c1)\,\alpha^2}{a1^3 + 2\,b1^3 - 6\,a1\,b1\,c1 + 4\,c1^3}$$

```
ai = na inv // Expand // Together
```

$$\left(a1^3 - 2\,a1\,b1\,c1 - 2\,b1^2\,c1\,\alpha + 2\,a1\,c1^2\,\alpha + b1^3\,\alpha^3 - 2\,a1\,b1\,c1\,\alpha^3 + 2\,c1^3\,\alpha^3 + b1^2\,c1\,\alpha^4 - a1\,c1^2\,\alpha^4\right) \Big/ \left(a1^3 + 2\,b1^3 - 6\,a1\,b1\,c1 + 4\,c1^3\right)$$

```
PolynomialRemainder[ai, pa, α] // Together
```

1

9.3 Solving Polynomial Equations

Solve[*poly* == 0, *var*]	Solves the algebraic equation in ***var*** by radicals if *Mathematica* can find such a solution.
NSolve[*poly* == 0, *var*]	Gets an approximate numeric solution to the algebraic equation in ***var***
NSolve[*poly* == 0, *var*, *n*]	Gets solutions to n-digit precision (but does not always succeed as wanted)
RootReduce[*expr*]	Attempt ro reduce ***expr*** to a single **Root** object

```
Solve[ x^6 == 1, x]
```

$\{\{x \to -1\}, \{x \to 1\}, \{x \to -(-1)^{1/3}\}, \{x \to (-1)^{1/3}\}, \{x \to -(-1)^{2/3}\}, \{x \to (-1)^{2/3}\}\}$

```
p = 3 + 3 x - 7 x^2 - x^3 + 2 x^4 + 3 x^7 - 3 x^8 - x^9 +
    x^10
```

$3 + 3 x - 7 x^2 - x^3 + 2 x^4 + 3 x^7 - 3 x^8 - x^9 + x^{10}$

```
Solve[p == 0, x]
```

{{x → 1}, {x → -√3}, {x → √3}, {x → Root[1 + 2 #1 + #1^7 &, 1]},
{x → Root[1 + 2 #1 + #1^7 &, 2]}, {x → Root[1 + 2 #1 + #1^7 &, 3]},
{x → Root[1 + 2 #1 + #1^7 &, 4]}, {x → Root[1 + 2 #1 + #1^7 &, 5]},
{x → Root[1 + 2 #1 + #1^7 &, 6]}, {x → Root[1 + 2 #1 + #1^7 &, 7]}}

Last entry -> running index of the roots; # ... unknown; the empersant symbol & closes the pure function.

```
N[%,30]
```

{{x → 1.00000000000000000000000000000}, {x → -1.73205080756887729352744634151},
{x → 1.73205080756887729352744634151}, {x → -0.496292071389336401136054258286},
{x → -0.868688301429947635782301355745 - 0.585281855209975056599758692485 i},
{x → -0.868688301429947635782301355745 + 0.585281855209975056599758692485 i},
{x → 0.0763555777486989843662590506097 - 1.14094614204992652592303958314 2 i},
{x → 0.0763555777486989843662590506097 + 1.14094614204992652592303958314 2 i},
{x → 1.04047875937591685198406943419 2 - 0.567349582684564970671772237342 i},
{x → 1.04047875937591685198406943419 2 + 0.567349582684564970671772237342 i}}

```
q = x^6 - 9 x^4 - 4 x^3 + 27 x^2 - 36 x - 23
```

$-23 - 36 x + 27 x^2 - 4 x^3 - 9 x^4 + x^6$

```
Solve[q == 0, x]
```

{{x → Root[-23 - 36 #1 + 27 #1^2 - 4 #1^3 - 9 #1^4 + #1^6 &, 1]},
{x → Root[-23 - 36 #1 + 27 #1^2 - 4 #1^3 - 9 #1^4 + #1^6 &, 2]},
{x → Root[-23 - 36 #1 + 27 #1^2 - 4 #1^3 - 9 #1^4 + #1^6 &, 3]},
{x → Root[-23 - 36 #1 + 27 #1^2 - 4 #1^3 - 9 #1^4 + #1^6 &, 4]},
{x → Root[-23 - 36 #1 + 27 #1^2 - 4 #1^3 - 9 #1^4 + #1^6 &, 5]},
{x → Root[-23 - 36 #1 + 27 #1^2 - 4 #1^3 - 9 #1^4 + #1^6 &, 6]}}

Even if a polynomial comprises higher powers than the fourth one, it may be possible to express its zeros by formulas involving root expressions. If *Mathematica* gives the answer in the Root form as above, this is no exact indication that zeros cannot be expressed by formulas involving roots. The sixth order polynomial **q** given above is such an example. x1 and x2 are two such zeros of this polynomia, which can be expressed by roots:

```
Factor[q]
```

$-23 - 36 x + 27 x^2 - 4 x^3 - 9 x^4 + x^6$

```
x1 = 2^(1/3) + 3^(1/2);
N[%]
```

2.99197

```
q /. x -> x1
```

$-23-36\left(2^{1/3}+\sqrt{3}\right)+27\left(2^{1/3}+\sqrt{3}\right)^2-4\left(2^{1/3}+\sqrt{3}\right)^3-9\left(2^{1/3}+\sqrt{3}\right)^4+\left(2^{1/3}+\sqrt{3}\right)^6$

```
ExpandAll[%]
```

0

```
RootReduce[%%]
```

0

```
x2 = 2^(1/3) - 3^(1/2);
N[%]
```

-0.47213

```
Factor[q]
```

$-23-36\,x+27\,x^2-4\,x^3-9\,x^4+x^6$

```
Factor[q, Extension → {x1, x2}]
```

$\left(2^{1/3}-\sqrt{3}-x\right)\left(2^{1/3}+\sqrt{3}-x\right)\left(3+2^{2/3}-2^{1/3}\sqrt{3}+\left(2^{1/3}-2\sqrt{3}\right)x+x^2\right)\left(3+2^{2/3}+2^{1/3}\sqrt{3}+\left(2^{1/3}+2\sqrt{3}\right)x+x^2\right)$

```
tr = Table[{x -> 2^(1/3) E^(I 2 Pi k/3) + Sqrt[3] (-1)^n},
    {k, 3}, {n, 2}]
```

$\left\{\left\{\left\{x\to-\sqrt{3}+2^{1/3}\,e^{\frac{2i\pi}{3}}\right\},\left\{x\to\sqrt{3}+2^{1/3}\,e^{\frac{2i\pi}{3}}\right\}\right\},\left\{\left\{x\to-\sqrt{3}+2^{1/3}\,e^{-\frac{2i\pi}{3}}\right\},\left\{x\to\sqrt{3}+2^{1/3}\,e^{-\frac{2i\pi}{3}}\right\}\right\},\left\{\left\{x\to2^{1/3}-\sqrt{3}\right\},\left\{x\to2^{1/3}+\sqrt{3}\right\}\right\}\right\}$

```
q /. tr // FullSimplify
```

{{0, 0}, {0, 0}, {0, 0}}

```
Solve[(Sqrt[-x] - a)^2 ==  b^2, x]
```

$\left\{\left\{x\to-a^2-2\,a\,b-b^2\right\},\left\{x\to-a^2+2\,a\,b-b^2\right\}\right\}$

```
Solve[(Sqrt[-x] - a)^2 == 1. * b^2, x]
```

$\left\{\left\{x\to0.5\left(-2.\,a^2-4.\,a\,b-2.\,b^2\right)\right\},\left\{x\to0.5\left(-2.\,a^2+4.\,a\,b-2.\,b^2\right)\right\}\right\}$

According to the theory of **Evariste Galois** the roots of every equation up to and including the 4th order can be expressed by radicals (= expressions involving only fractions and roots). For equations of higher order this is possible in exceptional cases only. Galois' theory based on group theory gives a means to investigate this question. *Mathematica* does not give such solutions even if this is possible in some of such cases. If *Mathematica* gives the answer in the Root form as above, this is no exact indication that zeros cannot be expressed by formulas involving roots. The sixth order polynomial **q** given above is such an example.

For some cubic equations all zeros are real, but is impossible to get exact root expressions without the imaginary unit i (casus irreducibilis). The polynomial below is such a case:

```
p = x^3 + x^2 - 20 x - 9;
```

```
Solve[p == 0, x]
```

$$\left\{\left\{x \to \frac{1}{3}\left(-1 + \frac{61^{2/3}}{\left(\frac{1}{2}\left(1 + 9 i \sqrt{3}\right)\right)^{1/3}} + \left(\frac{61}{2}\left(1 + 9 i \sqrt{3}\right)\right)^{1/3}\right)\right\},\right.$$
$$\left\{x \to -\frac{1}{3} - \frac{61^{2/3}\left(1 + i\sqrt{3}\right)}{3 \times 2^{2/3}\left(1 + 9 i \sqrt{3}\right)^{1/3}} - \frac{1}{6}\left(1 - i\sqrt{3}\right)\left(\frac{61}{2}\left(1 + 9 i \sqrt{3}\right)\right)^{1/3}\right\},$$
$$\left.\left\{x \to -\frac{1}{3} - \frac{61^{2/3}\left(1 - i\sqrt{3}\right)}{3 \times 2^{2/3}\left(1 + 9 i \sqrt{3}\right)^{1/3}} - \frac{1}{6}\left(1 + i\sqrt{3}\right)\left(\frac{61}{2}\left(1 + 9 i \sqrt{3}\right)\right)^{1/3}\right\}\right\}$$

```
N[%] // Chop
```

{{x → 4.23048}, {x → -4.78597}, {x → -0.444512}}

```
ComplexExpand[x /. %]
```

{4.23048, -4.78597, -0.444512}

The above expressions are throughout real, but they contain the imaginary unit i necessarily.

```
eq = a x^4 + b x^3 + c x^2 + d x + e == 0
```

$$1 + 6 x + d x + 9 x^2 + c x^2 + b x^3 + a x^4 + 8 y^2 + 24 x y^2 + 16 y^4 == 0$$

```
Solve[eq, x] ; (* The output fills 5 pages *)
```

RootReduce[*expr*]	Attempt to reduce ***expr*** to a single **Root** object
ToRadicals[*expr*]	Attempt to transform **Root** objects to explicit radicals

```
p = x^3 + 2 x - 2
```

$-2+2x+x^3$

```
so = Solve[p == 0]
```

$$\left\{\left\{x\to\frac{\left(9+\sqrt{105}\right)^{1/3}}{3^{2/3}}-\frac{2}{\left(3\left(9+\sqrt{105}\right)\right)^{1/3}}\right\},\right.$$
$$\left\{x\to-\frac{\left(1+i\sqrt{3}\right)\left(9+\sqrt{105}\right)^{1/3}}{2\times 3^{2/3}}+\frac{1-i\sqrt{3}}{\left(3\left(9+\sqrt{105}\right)\right)^{1/3}}\right\},$$
$$\left.\left\{x\to-\frac{\left(1-i\sqrt{3}\right)\left(9+\sqrt{105}\right)^{1/3}}{2\times 3^{2/3}}+\frac{1+i\sqrt{3}}{\left(3\left(9+\sqrt{105}\right)\right)^{1/3}}\right\}\right\}$$

```
(x /. so[[2]]) (x /. so[[3]])
```

$$\left(-\frac{\left(1+i\sqrt{3}\right)\left(9+\sqrt{105}\right)^{1/3}}{2\times 3^{2/3}}+\frac{1-i\sqrt{3}}{\left(3\left(9+\sqrt{105}\right)\right)^{1/3}}\right)$$
$$\left(-\frac{\left(1-i\sqrt{3}\right)\left(9+\sqrt{105}\right)^{1/3}}{2\times 3^{2/3}}+\frac{1+i\sqrt{3}}{\left(3\left(9+\sqrt{105}\right)\right)^{1/3}}\right)$$

```
RootReduce[%]
```

Root$\left[-4-2\,\#1^2+\#1^3\ \&,\ 1\right]$

```
ToRadicals[%]
```

$$\frac{1}{3}\left(2+\left(62-6\sqrt{105}\right)^{1/3}+\left(2\left(31+3\sqrt{105}\right)\right)^{1/3}\right)$$

SolveAlways[*eq*, {*x*,*y*}]	gives the values of the parameters that make the equation ***eq*** valid for all values of the variables ***x*** and ***y***

```
Clear[a, b, x, y, r]
eq = (x^2 + 12 x + y^2 - 20 y + 15 == (x - a)^2 + (y - b)^2 - r^2);
Solve[eq, {a, b, r}]
```

Solve::svars: Equations may not give solutions for all "solve" variables. ≫

$$\left\{\left\{r\to-\sqrt{-15+a^2+b^2-12x-2ax+20y-2by}\right\},\left\{r\to\sqrt{-15+a^2+b^2-12x-2ax+20y-2by}\right\}\right\}$$

Solve::svars : "Equations may not give solutions for all " solve " variables.

$$\left\{\left\{r\to-\sqrt{-15+a^2+b^2-12x-2ax+20y-2by}\right\},\left\{r\to\sqrt{-15+a^2+b^2-12x-2ax+20y-2by}\right\}\right\}$$

Above **eq** has been regarded as an equation for the single unknown **r.** One may look at **eq** in a different way: There are two different expressions on the left-hand and on the right-hand side. On asks: for which values of a, b, r are the sides identical functions in x and y ?

```
soa = SolveAlways[eq, {x, y}]
```

{{a → -6, b → 10, r → -11}, {a → -6, b → 10, r → 11}}

```
eq /. soa
```

$\{15 + 12 x + x^2 - 20 y + y^2 == -121 + (6 + x)^2 + (-10 + y)^2,$
$15 + 12 x + x^2 - 20 y + y^2 == -121 + (6 + x)^2 + (-10 + y)^2\}$

```
FullSimplify[%]
```

{True, True}

```
?? Eliminate
```

Eliminate[*eqns*, *vars*] eliminates variables between a set of simultaneous equations >>

Attributes[Eliminate] = {Protected}

Options[Eliminate] = {InverseFunctions → Automatic, MakeRules → False, Method → 1, Mode → Generic, Sort → True, VerifySolutions → Automatic, WorkingPrecision → ∞}

```
eqs = x^2 + y^2 + z^2 == 1 && x - y + z == 2 && x^3 - y^2 == z + 1
```

$x^2 + y^2 + z^2 == 1 \&\& x - y + z == 2 \&\& x^3 - y^2 == 1 + z$

```
Eliminate[eqs, z]
```

$-18 x + 4 x^2 - 28 x^3 + 8 x^4 + 4 x^5 + 4 x^6 == -27 \&\& y == -12 - 2 x - 5 x^2 + 8 x^3 + 4 x^4 + 2 x^5$

```
Eliminate[eqs, {y, z}]
```

$-18 x + 4 x^2 - 28 x^3 + 8 x^4 + 4 x^5 + 4 x^6 == -27$

```
Eliminate[Sqrt[x] + Sqrt[y - 1] == 1 && 2 x^(1 / 3) + 3 y^2 == 2, y]
```

$-2\,864\,000 x + 4\,956\,544 x^2 - 3\,109\,824 x^3 + 4\,114\,800 x^4 - 1\,219\,104 x^5 +$
$819\,936 x^6 - 310\,608 x^7 + 121\,500 x^8 - 23\,328 x^9 + 14\,580 x^{10} + 729 x^{12} == -1\,000\,000$

9.3.1 Solving Polynomial Equations. History and Further Developements

Any polynomial equation up to the 4 th degree with real or complex coefficients

$$p_4(x) := a_4\, x^4 + a_3\, x^3 + a_2\, x^2 + a_1\, x + a_0 \;=\; 0 \qquad (1)$$

can be solved by radicals; i.e. it can by solved by addition, subtraction, multiplication, division and root extraction.This is acomplished by *Mathematica* by the command **Solve[]** as long as eq.(1) does not contain any decimal numbers.
Of course, any polynomial equation has as many roots as its degree (counting multiplicities). But these can be expressed in any case by the five operations just listed only up to poynomials of the fourth degree. This is the content of Abel's impossibility theorem. Some special polynomial equations of degree 5 or higher may still be solvable by radicals. It is no proof to the contrary if **Solve[]** does not succeed in finding such a solution of an equation. The theory of Galois is a means to check whether an equation is solvable by radicals.

The solution theory of the polynomial equation of fifth degree was developed by Brioschi, Hermite and Kronecker.
A comprehensive representation of this work was given by F. Klein. A reedition of Klein’s book by Slodowy contains an introduction, a decription of further work and a list with more recent references.

F, Klein: Vorlesungen ueber das Ikosaeder und die Auflösung der Gleichung fünften Grades. Birkhäuserverlag, Basel, 1993. Neuausgabe des 1884 erschienenen Werkes, hier mit einem umfangreichen Vorwort und nachstehendem Kommentar von P. Slodowy.

The historical developement of the solution theory is listed in the notes M9EquationsHistory.doc.

9.3.2 Solving Inequalities

```
Reduce[Abs[-3 + x] <= (x + 1) Abs[4 + x], x, Reals]
```

$x \geq -3 + 2\sqrt{2}$

```
v0 = Last[%] // N
```

-0.171573

```
Plot[-Abs[-3 + x] + (x + 1) Abs[4 + x],
 {x, -5, 1}, PlotRange → {-9, 4}, AspectRatio → 0.3`,
 Epilog → {Thickness[0.01`], Hue[0], Line[{{v0, 0}, {1.1`, 0}}]}]
```




```
Reduce[x (x^2 - 2) (x^2 - 3) > 0, x, Reals]
```

$-\sqrt{3} < x < -\sqrt{2}\ ||\ 0 < x < \sqrt{2}\ ||\ x > \sqrt{3}$

```
Plot[x (x^2 - 2) (x^2 - 3), {x, -2, 2}, PlotRange → 3 {-1, 1}, AspectRatio → 0.4`,
 Epilog → {Thickness[0.01`], Hue[0], Line[{{-√3, 0}, {-√2, 0}}],
   Line[{{0, 0}, {√2, 0}}], Line[{{√3, 0}, {2.1`, 0}}]}]
```



In the example below the unknowns are restricted to

```
Clear[x1, x2, x3]
tx = {x1,x2,x3^2};
```

```
ma = {{2,1,3},{4,3,5}}; MatrixForm[ma]
```

$$\begin{pmatrix} 2 & 1 & 3 \\ 4 & 3 & 5 \end{pmatrix}$$

```
ma.tx <= {7,9}//Thread;
sys = {ungl1,ungl2} = %
```

$\{2\,x1 + x2 + 3\,x3^2 \le 7,\ 4\,x1 + 3\,x2 + 5\,x3^2 \le 9\}$

```
Reduce[Join[sys,
{(x1==0||x1==1),(x2==0||x2==1),(x3==0||x3==1)}],
                {x1,x2,x3}]
```

```
(x1 == 0 && x2 == 0 && x3 == 0) || (x1 == 0 && x2 == 0 && x3 == 1) ||
 (x1 == 0 && x2 == 1 && x3 == 0) || (x1 == 0 && x2 == 1 && x3 == 1) ||
 (x1 == 1 && x2 == 0 && x3 == 0) || (x1 == 1 && x2 == 0 && x3 == 1) || (x1 == 1 && x2 == 1 && x3 == 0)
```

Tnis result is translated into sets of substitions:

```
sos = {{x1 -> 0, x2 -> 0, x3 -> 0},
       {x1 -> 0, x2 -> 0, x3 -> 1},
       {x1 -> 0, x2 -> 1, x3 -> 0},
       {x1 -> 0, x2 -> 1, x3 -> 1},
       {x1 -> 1, x2 -> 0, x3 -> 0},
       {x1 -> 1, x2 -> 0, x3 -> 1},
       {x1 -> 1, x2 -> 1, x3 -> 0}};
```

```
sys /. sos
```

```
{{True, True}, {True, True}, {True, True},
 {True, True}, {True, True}, {True, True}, {True, True}}
```

9.3.2 Solving Transcendental Equations

```
Solve[Cos[x] == 0, x]
```

$\{\{x \to \text{ConditionalExpression}[-\frac{\pi}{2} + 2\,\pi\,C[1],\ C[1] \in \text{Integers}]\},$
$\{x \to \text{ConditionalExpression}[\frac{\pi}{2} + 2\,\pi\,C[1],\ C[1] \in \text{Integers}]\}\}$

Solve::"ifun" :
Inverse functions are being used by Solve, so some solutions may not be found;
 use Reduce for complete solution information.

```
Reduce[Cos[x] == 0, x]
```

$C[1] \in \text{Integers}\ \&\&\ \left(x == -\frac{\pi}{2} + 2\,\pi\,C[1]\ ||\ x == \frac{\pi}{2} + 2\,\pi\,C[1]\right)$

This gives all roots.

```
Solve[Sin[x] + 3 Cos[x] == 0, x]
```

```
{{x → ConditionalExpression[-ArcTan[3] + 2 π C[1], C[1] ∈ Integers]},
 {x → ConditionalExpression[π - ArcTan[3] + 2 π C[1], C[1] ∈ Integers]}}
```

Solve::"ifun" :
Inverse functions are being used by Solve, so some solutions may not be foundÖ
 use Reduce for complete solution information.

```
N[%]
```

```
{{x → ConditionalExpression[-1.24905 + 6.28319 C[1], C[1] ∈ Integers]},
 {x → ConditionalExpression[1.89255 + 6.28319 C[1], C[1] ∈ Integers]}}
```

```
Reduce[Sin[x] + 3 Cos[x] == 0, x]
```

$C[1] \in \text{Integers}$ &&
$\left(x == 2\,\text{ArcTan}\left[\frac{1}{3}\left(1-\sqrt{10}\right)\right] + 2\,\pi\,C[1] \;||\; x == 2\,\text{ArcTan}\left[\frac{1}{3}\left(1+\sqrt{10}\right)\right] + 2\,\pi\,C[1]\right)$

This gives all roots. These consist of either one of the basis expressions given below plus 2π x an integer.

$\text{N}\left[\left\{2\,\text{ArcTan}\left[\frac{1}{3}\left(1+\sqrt{10}\right)\right],\ 2\,\text{ArcTan}\left[\frac{1}{3}\left(1-\sqrt{10}\right)\right]\right\}\right]$

```
{1.89255, -1.24905}
```

9.4 Reduce

```
?? Reduce
```

Reduce[*expr*, *vars*] reduces the statement *expr* by solving equations or inequalities for *vars* and eliminating quantifiers. Reduce[*expr*, *vars*, *dom*] does the reduction over the domain *dom*. Common choices of *dom* are Reals, Integers and Complexes >>

```
Attributes[Reduce] = {Protected}

Options[Reduce] = {Backsubstitution → False, Cubics → False, GeneratedParameters → C,
  Method → Automatic, Modulus → 0, Quartics → False, WorkingPrecision → ∞}
```

lhs == *rhs*	equations
lhs != *rhs*	inequations
lhs > *rhs* or *lhs* >= *rhs*	inequalities
expr ∈ *dom*	domain specifications
ForAll[*x*, *cond*, *expr*]	universal quantifiers
Exists[*x*, *cond*, *expr*]	existential quantifiers

```
Clear[n, q, x, y]

Reduce[x^2 + y^2 < 1, {x, y}]
```

$-1 < x < 1 \; \&\& \; -\sqrt{1 - x^2} < y < \sqrt{1 - x^2}$

```
Reduce[x^2 - 7 y^2 == 1 && x > 0 && y > 0, {x, y}, Reals]
```

$x > 1 \; \&\& \; y == \frac{\sqrt{-1 + x^2}}{\sqrt{7}}$

```
Reduce[Exists[{x, y}, x^2 + a y^2 ≤ 1 && x - y ≥ 2], a]
```

$a \le \frac{1}{3}$

```
Reduce[2 x + 3 y - 5 z == 1 && 3 x - 4 y + 7 z == 3, {x, y, z}, Reals]
```

$y == 22 - 29\,x \; \&\& \; z == 13 - 17\,x$

```
Reduce[Abs[3 x^2 - 7 x - 6] < Abs[x^2 + x], x, Reals]
```

$\frac{1}{4}\left(3 - \sqrt{33}\right) < x < 2 - \sqrt{7} \; || \; \frac{1}{4}\left(3 + \sqrt{33}\right) < x < 2 + \sqrt{7}$

```
Reduce[BesselJ[n - 1, q] - 2 n / q BesselJ[n, q] + BesselJ[n + 1, q] == 0, {q}]
```

Reduce::fexp:
Warning: Reduce used FunctionExpand to transform the system. Since FunctionExpand transformation rules are only generically correct, the solution set might have been altered. >>

$q \neq 0$

Reduce::fexp: “Warning: Reduce used FunctionExpand to transform the system. Since FunctionExpand transformation rules
are only generically correct, the solution set might have been altered.”

```
Limit[BesselJ[n - 1, q] - 2 n / q BesselJ[n, q] + BesselJ[n + 1, q], {q → 0}]
```

{0}

9.5 Finding Approximate Roots of Expressions by Iterative Methods

9.5.1 One Unknown

FindRoot[exp,{x,x_0}] searches for a root of the algebraic or transcendental expression **exp** starting the iteration with **x = x_0**.

FindRoot[exp,{x,x_0,x_1}]	if *Mathematica* cannot find the derivative of **exp** then it needs two starting values **x_0,x_1** to work with secants

This works also for complex expressions. It finds complex roots of real expressions.

```
FindRoot[Sin[x] + 3 Cos[x] , {x, 2.1} ]
```

```
{x → 1.89255}
```

```
f = z^2 + Conjugate[z] + 2 I
```

```
2 ⅈ + z^2 + Conjugate[z]
```

```
FindRoot[f, {z, 1 + 2 I}]
```

FindRoot::lstol :
The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances. ≫

```
{z → 0.0979449 + 0.498099 ⅈ}
```

```
f /. %
```

```
-0.140564 + 1.59947 ⅈ
```

This shows that it is important to find a good starting value. Therefore we plot the complex function:

```
Plot3D[Abs[f /.z → x + ⅈ y], {x, -3, 3}, {y, -3, 3}]
```




```
FindRoot[f, {z, 2 - 2 I}]
```

```
{z → 1.14017 - 1.5621 ⅈ}
```

```
f /. %
```

$0. - 7.7915 \times 10^{-9}\, i$

```
FindRoot[f, {z, - 1}]
```

FindRoot::cvmit : Failed to converge to the requested accuracy or precision within 100 iterations. ≫

```
{z → -1.25747 + 0.569 ⅈ}
```

```
f /. %
```

$0. - 2.30401 \times 10^{-8}\, i$

One may give also two starting vales. In some cases FindRoot[] even asks for this.

```
Clear[x, fx]
```

```
fx = Sin[x] + x - 2;
```

```
p1 = Plot[ⅇ^Abs[x] - 2, {x, -1, 1}, PlotLabel → "ⅇ^Abs[x] - 2\n"];
p2 = Plot[fx, {x, 0, 2}, PlotLabel → fx];
Show[GraphicsRow[{p1, p2}], ImageSize → 400]
```




```
FindRoot[ⅇ^Abs[x] == 2, {x, 1}]
```

{x → 0.693147}

9.4.1.1 Roots with Higher Precision

```
FindRoot[fx, {x, 1}, MaxIterations → 50,
    AccuracyGoal → 24, WorkingPrecision → 34]
```

{x → 1.106060157706271910616737297030074}

```
fx /. %
```

$0. \times 10^{-34}$

9.5.2 n Expressions for n Unknowns

```
Solve[{x^3 - y^2 == 0, x + y == 2}, {x, y}]
```

{{x → -2 i, y → 2 + 2 i}, {x → 2 i, y → 2 - 2 i}, {x → 1, y → 1}}

```
ContourPlot[{x^3 - y^2 == 0, x + y == 2}, {x, 0, 2}, {y, 0, 2}]
```




```
FindRoot[{x^3 - y^2, x + y - 2}, {x, 1}, {y, 2}]
```

{x → 1., y → 1.}

```
FindRoot[{x^3 - y^2, x + y - 2}, {x, I}, {y, 2 I}]
```

{x → 3.6393×10^{-17} - 2. i, y → 2. + 2. i}

```
Chop[%]
```

{x → 0. - 2. i, y → 2. + 2. i}

```
FindRoot[{x^3 - y^2, x + y - 2}, {x, I}, {y, -2 I}]
```

{x → -2.08811×10^{-17} + 2. i, y → 2. - 2. i}

```
Chop[%]
```

{x → 0. + 2. i, y → 2. - 2. i}

9.5.2.1 Roots with Higher Precision

```
FindRoot[{x^3 - 2 y^2, x + y - 2}, {x, I}, {y, -2 I},
 AccuracyGoal → 24, WorkingPrecision → 34]
```

{x →
 0.4301597090019467340886000418804314 + 2.614282557364090960984705147027531 i,
 y → 1.569840290998053265911399958119569 - 2.614282557364090960984705147027531 i}

9.6 Exercises

9.1. Divide the polynomial f by the polynomial g. Find the quotient and the remainder.

1) $f = x^6 - 5x^5 + 4x^4 - 2x^3 + 3x^2 - x - 1$, $g = x^2 - x + 1$;
2) $f = x^7 + x^6 - 1$, $g = x - 1$.

9.2 A square of given side length s is inscribed into a right triangle such that one of its corners coincides with the right apex and two of its sides are aligned with the legs of the triangle. The length of the hypotenuse, c, is given.
a) Determine the lengths of the legs.
b) Specialize to s = 4, c = 20. Draw the solutions, together with the square filled. .

9.3 Compute f(a,b,c) = a^4 + b^4 + c^4, where
a + b + c = 3, a^2 + b^2 + c^2 = 9, a^3 + b^3 + c^3 = 24.
The result f(a,b,c) is an integer.

9.4 For which real value(s) of the parameter **a** occur multiple roots of the following equation :
$$a^4 + 3a^2 x + a x^4 + x^5 = 0$$
These value(s) should be given with an accuracy of 30 decimal places.